
// LastFour.WCM — open the docs listed at the bottom of the File menu.
//
// Author:
// Richard Reiner <rreiner@nexus.yorku.ca>
//
// Revision history:
// 3 Jul 93 RjR First version.
// 13 Jul 93 Paul H. Caron Added 'FileAssign@' and 'SelectFiles@'.
// 13 Jul 93 RjR Arranged not to show 4 entries in the 
SelectFiles dialogue if there are fewer than 4 last-opened entries in the INI file; tuned the 
look of the dialogue; improved error trapping.

APPLICATION(WP;WPWP;Default)

//************************************************************
// Main program
OnError (End@)
OnCancel (End@)
OnNotFound Call(NotFound@)
Call(LoadDLL@)
Call(ReadINI@)
Call(SelectFiles@)
Call(OpenFiles@)
Call(End@)

//************************************************************
// Subr ReadINI@
Label(ReadINI@)

ItemCount:=0
ItemId:=0
Entries:=""
Tag:=""
TagLength:=0
IniResult:=0
FileName:=""
FileName1:=""
FileName2:=""
FileName3:=""
FileName4:=""

// Find out how many entries there are in the LastOpened section of WPWP.INI
DLLCall(MCRLink; 

"MCRQueryIniSection"; 
ItemCount : WORD;
{ AnsiString("WPWP");

AnsiString("LastOpened");
Address(Entries)})

// Retrieve the data and store it
For(ItemId;1;ItemId<=ItemCount;ItemId+1)

DLLCall(MCRLink;
"MCRQueryIniTag";
TagLength : WORD;
{ Entries;

Address(Tag);



LOWORD(ItemID)
})

DLLCall(MCRLink; 
"MCRGetIniString"; 
IniResult : WORD;
{ "WPWP";

AnsiString("LastOpened");
AnsiString(Tag);
AnsiString("Error");
Address(FileName)

})
Call (FileAssign@)

Endfor
Return

//*********************************************************
// Subr FileAssign@
Label(FileAssign@)
If (FileName1="")

FileName1:=FileName
Return

Else
If (FileName2="")

FileName2:=FileName
Return

Else
If(FileName3="")

FileName3:=FileName
Return

Else
FileName4:=FileName
Return

EndIf
EndIf

EndIf

//************************************************************
// Subr SelectFiles@
// Query the user about which files to open, and open them.
Label(SelectFiles@)

ChkBx1:=1
ChkBx2:=1
ChkBx3:=1
ChkBx4:=1

DialogDefine(100; 50; 50; 200; 135; 1+2+16; "LastFour.WCM")

If (FileName1 <> "")
DialogAddCheckBox(100; 2000; 16; 20; 170; 15; "&1 - "+FileName1; ChkBx1)

EndIf
If (FileName2 <> "")

DialogAddCheckBox(100; 2001; 16; 35; 170; 15; "&2 - "+FileName2; ChkBx2)
EndIf
If(FileName3 <> "")



DialogAddCheckBox(100; 2002; 16; 50; 170; 15; "&3 - "+FileName3; ChkBx3)
EndIf
If(FileName4 <> "")

DialogAddCheckBox(100; 2003; 16; 65; 170; 15; "&4 - "+FileName4; ChkBx4)
EndIf

DialogAddGroupBox(100; 100; 8; 6; 180; 82; "Select files to re-open")

DialogDisplay(100; 2000)
Result:=MacroDialogResult
DialogDestroy(100)
If (Result=2)

Quit
EndIf

Return

//************************************************************
// Subr OpenFiles@
// Open selected files.  NotFound@ expects the current file to be in variable FileName.
Label(OpenFiles@)

If ((ChkBx1 = 1) and (Filename1 <> ""))
FileName:=FileName1
FileOpen(FileName)

EndIf
If ((ChkBx2 = 1) and (Filename2 <> ""))

FileName:=FileName2
FileOpen(FileName)

EndIf
If ((ChkBx3 = 1) and (Filename3 <> ""))

FileName:=FileName3
FileOpen(FileName)

EndIf
If ((ChkBx4 = 1) and (Filename4 <> ""))

FileName:=FileName4
FileOpen(FileName)

EndIf
Return

//************************************************************
// Subr LoadDLL@
Label(LoadDLL@)

DllLoad(MCRLink;"WPCMCR.DLL")
Return

//************************************************************
// Subr NotFound@
Label(NotFound@)

Beep
Prompt("LastFour.WCM"; "Warning: the file """+FileName+""" was not found."; 4 ; ;)
Pause
Endprompt

Return

//************************************************************



// Subr End@
Label(End@)

DllFree(MCRLink)
Quit

Return


